Journal of Chromatography, 131 (1977) 121–130

CHROM. 9511

COMPUTER PROCESSING OF FATTY ACID ANALYSIS DATA

J. W. ASTON

Otto Madsen Dairy Research Laboratory, Hercules Street, Hamilton, Queensland 4007 (Australia)

(Received May 18th, 1976)

SUMMARY

A Fortran computer program for the processing of fatty acid data from the analysis of fats and oils by gas–liquid chromatography is described. The analytical method and calculations are described primarily for the analysis of butterfat and margarine fat, but with minor changes could be adapted to suit other fats and oils. The program computes the concentration of each fatty acid as weight % and mole %, the percentage glycerol, the theoretical iodine value, and other relevant combinations of fatty acids.

INTRODUCTION

Not too many years ago, the number of samples that could be analysed in the aboratory was limited principally by the speed of the analytical technique employed. However, with the advent of gas–liquid chromatographic (GLC) techniques, the number of analyses able to be performed increased greatly and became, in many cases, limited only by the speed at which the results could be processed. In recent years, use of the computer has tended to reverse this situation by its ability to process large volumes of data in extremely short times. Today, by combining GLC with computer data processing, a far greater number of analyses can be performed than ever before.

The GLC analysis of fats and oils for their fatty acid compositions is a routine analytical procedure in many laboratories today. For classification and identification of fats and oils, usually only measurement of the major even-numbered carbon fatty acids is required, while for more demanding compositional and stereochemical studies, the odd-numbered carbon and branched-chain acids are often required. A typical chromatogram may contain up to 30 peaks and the calculations, being involved and tedious, are ideally suited to computer handling. Numerous methods for fatty acid analysis have been published covering a wide variety of applications[1–7]. However, little information has appeared in the literature regarding the calculations involved and the computer processing of the results. For this reason, this paper describes a Fortran program to perform the required calculations adapted for the method used in this laboratory.

EXPERIMENTAL

Equipment

A Varian Aerograph Model 1525 gas chromatograph was used fitted with a 2.4 m × 3 mm I.D. borosilicate glass column packed with 15% EGSS-X on 100–120 mesh Gas-Chrom P (Applied Science Labs., State College, Pa., U.S.A.), and a hydrogen flame ionisation detector. Further an Autolab Type 6300 electronic digital integrator and a Hitachi Type QPD54 potentiometric recorder were employed.

Method of analysis

Transesterification of the triglyceride fatty acids was carried out by the method of Christopherson and Glass[8] with the exception that 2 *N* KOH in *n*-butanol was used to produce the butyl esters. The reaction was carried out in plastic-capped centrifuge tubes which, after standing for 5 min at room temperature, were centrifuged at 2250 × *g* (av.) for 5 min. Of the clear upper portion 2 μl were injected into the gas chromatograph.

Operating conditions

The operating conditions used were as follows: carrier gas (nitrogen) flow-rate, 35 ml/min; hydrogen flow-rate, 30 ml/min; air flow-rate, 300 ml/min; detector temperature, 250°; injector temperature, 200°; initial column temperature, 90°; final column temperature, 225°; temperature program rate, 6°/min from 90–225°, then isothermal for 18 min.

CALCULATIONS

Response factors

As each fatty acid (f.a.) responds differently to the flame ionisation detector, a response correction factor should be applied. These are predetermined by the analysis of a standard mixture of triglycerides (trig.) of the required fatty acids. The response factors are calculated relative to myristic acid (C14:0), which is arbitrarily assigned a value of 1.0. The calculations are as follows

$$\text{Weight response factor} = \frac{\text{Area of f.a.}}{\text{Weight of f.a. in trig.}} \times \frac{\text{Weight of C14:0 in trig.}}{\text{Area of C14:0}}$$

$$\text{Molar response factor} = \text{Weight response factor} \times \frac{\text{Mol. wt. of f.a.}}{\text{Mol. wt. of C14:0}}$$

The peak areas are then corrected by dividing each by its response factor

$$\text{Weight corrected area of f.a.} = \frac{\text{Area of f.a.}}{\text{Weight response factor}}$$

$$\text{Mole corrected area of f.a.} = \frac{\text{Area of f.a.}}{\text{Molar response factor}}$$

Any unidentified peaks on the chromatogram are assigned a response factor of 1.0.

Fatty acid concentrations

The calculations of the fatty acid concentrations are carried out in two sections. The first is for acids for which response factors are available. These are the major peaks. The second section consists of the peaks whose identity is unknown and which have therefore been assigned response factors of 1.0. These are the minor peaks and are grouped together to give the total minor peak concentration. The total corrected area is the sum of the corrected areas of the major peaks plus the areas of the minor peaks. The concentration of each major peak is expressed as weight % and mole % as follows

$$\text{Weight } \% = \frac{\text{Weight corrected area of f.a.}}{\text{Total weight corrected area}} \times 100$$

$$\text{Mole } \% = \frac{\text{Mole corrected area of f.a.}}{\text{Total mole corrected area}} \times 100$$

Percentage of glycerol

The percentage of glycerol is required for the calculation of the theoretical iodine value. The procedure is as follows

(i) Assume 100 g of total fatty acids.

(ii) Convert the weight of each acid to moles by dividing each weight % by its molecular weight. (For the minor peaks, a molecular weight equal to that of C14:0 is assumed.)

(iii) Sum to calculate the total moles of fatty acids.

(iv) Divide by 3 to calculate the number of moles of glycerol associated with 100 g of fatty acids.

(v) Multiply by the molecular weight of glycerol to find the weight of glycerol associated with 100 g of fatty acids.

(vi) Then weight % glycerol

$$= \frac{\text{Weight of glycerol associated with 100 g of f.a.'s}}{100 + \text{weight of glycerol associated with 100 g of f.a.'s}} \times 100$$

Theoretical iodine value

The theoretical iodine value is useful for comparing estimates of unsaturation obtained from GC profiles and published experimental iodine values, especially when only the latter is available from the literature. It is expressed as the weight of iodine that reacts with the double bonds of the major unsaturated fatty acids in 100 g of fat or oil. The calculation is as follows

(i) Assume 100 g of total fatty acids.

(ii) Calculate the iodine mole equivalent of the unsaturated fatty acids as follows:

$$\text{Iodine mole equivalent} = \sum\left(\frac{\text{Wt.\% of unsaturated f.a.}}{\text{Mol. Wt. of f.a.}} \times \text{No. of double bonds} \times 2\right)$$

(iii) Multiply by the atomic weight of iodine to find the weight of iodine associated with 100 g of fatty acids.

(iv) Then theoretical iodine value

$$= \frac{\text{Weight of iodine associated with 100 g of f.a.'s}}{100 + \text{weight of glycerol associated with 100 g of f.a.'s}} \times 100$$

COMPUTER PROGRAM

The program was written in Fortran IV for an I.C.L. 1904A computer. The hardware requirements are 5K Core store, 1 line printer and 1 card punch.

Program description

The program calculates and/or prints out the following information:

1. Number of minor peaks
2. Uncorrected areas
3. Weight response factors
4. Areas corrected by weight response factors
5. Concentration in weight %
6. Mole response factors
7. Areas corrected by mole response factors
8. Concentration in mole %
9. % Glycerol
10. Theoretical iodine value

This information is calculated for the following fatty acids and combinations of fatty acids:

1. Butyric	C4:0	12. Stearic	C18:0
2. Caproic	C6:0	13. Oleic	C18:1
3. Caprylic	C8:0	14. Linoleic	C18:2
4. Capric	C10:0	15. Linolenic	C18:3
5. Lauric	C12:0	16. Nonadecanoic	C19:0
6. Myristic	C14:0	17. Arachidic	C20:0
7. Myristoleic	C14:1	18. Major peaks	
8. Pentadecanoic	C15:0	19. Minor peaks	
9. Palmitic	C16:0	20. Total	
10. Palmitoleic	C16:1	21. Total saturated acids	
11. Heptadecanoic	C17:0	22. Total unsaturated acids	
		23. *cis*-Methylene-interrupted unsaturated acids	

Data card details

The input data must be punched on cards in the following order:

(1) The header card contains the number of analyses to be computed in a particular run. The number must be entered right justified in the first two columns. This number cannot exceed 99.

(2) The second card contains the weight response factors for the seventeen major fatty acids. These numbers should fit on two cards with one space between consecutive numbers. The response factors must be predetermined by analysis of a standard mixture of the appropriate triglycerides. The mole response factors are calculated by the program.

(3) The next card contains the number of minor peaks appearing in the first chromatogram. The minor peaks are all peaks appearing in the chromatogram excluding the solvent/s peak and the major fatty acid peaks (C4:0 to C20:0). The number of minor peaks is punched right justified in columns 1 and 2. The next 70

columns (*i.e.*, from column 3 to column 72) on this card may be used for a sample description text.

(4) The next card contains the areas of the seventeen major peaks. These areas must be punched in the order shown in the program description above and a value for each acid must be entered. If the acid does not appear on the chromatogram, the value 0.0 is punched. The seventeen values may occupy two cards.

(5) The next card contains the areas of the minor peaks. The number of values punched must correspond to the figure on card 3. If there are no minor peaks the value 0.0 must be punched.

(6) The data in 3, 4, and 5 are then repeated for the second and for subsequent chromatograms.

Example

1. 07
2. 1.35 1.22 1.12 1.07 1.01 1.00 1.00 1.00 0.98 0.96 0.97 0.96 0.94 0.79 0.62 0.96 0.94
3. 06 MEDIUM FRACTION TRIGLYCERIDE BAND 9/1/76
4. 15000 7115 3306 8273 11723 47577 4540 8895 137683 7755 8988 101459 75606 1053 0.0 3629 600
5. 1531 4054 3557 778 2235 552
6. 04 LIGHT FRACTION TRIGLYCERIDE BAND 9/1/76
7. 3000 1491 709 3611 6833 46564 2622 12449 177991 7108 12770 148776 69259 1000 0.0 6218 1281
8. 1736 4273 3958 3054
9. etc. until 7 sets of data are punched.

Program

```
      MASTER FATA ANALYSIS
      DIMENSION ANAME(2,17),GFW(17),WR(17),SR(17),X(17),CORW(17),CORM(17
     1),Y(20),WTPC(17),SLPC(17),D(9)
      DATA ANAME(1,1)/16HBUTYRIC     C 4.0/,GFW(1)/88.1/
      DATA ANAME(1,2)/16HCAPROIC     C 6.0/,GFW(2)/116.16/
      DATA ANAME(1,3)/16HCAPRYLIC    C 8.0/,GFW(3)/144.21/
      DATA ANAME(1,4)/16HCAPRIC      C10.0/,GFW(4)/172.27/
      DATA ANAME(1,5)/16HLAURIC      C12.0/,GFW(5)/200.3/
      DATA ANAME(1,6)/16HMYRISTIC    C14.0/,GFW(6)/228.35/
      DATA ANAME(1,7)/16HMYRISTOLEI  C14.1/,GFW(7)/226.35/
      DATA ANAME(1,8)/16HPENTADECAN  C15.0/,GFW(8)/242.38/
      DATA ANAME(1,9)/16HPALMITIC    C16.0/,GFW(9)/256.41/
      DATA ANAME(1,10)/16HPALMITOLEI C16.1/,GFW(10)/254.39/
      DATA ANAME(1,11)/16HMARGARIC   C17.0/,GFW(11)/270.43/
      DATA ANAME(1,12)/16HSTEARIC    C18.0/,GFW(12)/284.46/
      DATA ANAME(1,13)/16HOLEIC      C18.1/,GFW(13)/282.44/
      DATA ANAME(1,14)/16HLINOLEIC   C18.2/,GFW(14)/280.43/
      DATA ANAME(1,15)/16HLINOLENIC  C18.3/,GFW(15)/278.41/
      DATA ANAME(1,16)/16HNONADECANO C19.0/,GFW(16)/298.50/
      DATA ANAME(1,17)/16HARACHIDIC  C20.0/,GFW(17)/312.51/
C
C LOGICAL UNIT 5 = CARD READER
C LOGICAL UNIT 2 = LINE PRINTER
C
      READ(5,1)M
    1 FORMAT(I2)
C
C M=NO OF ANALYSES
C
```

```
      READ(5,2)(WR(I),I=1,17)
    2 FORMAT(17F0.0)
C
C WR=WEIGHT RESPONSE FACTOR
C
      DO 3 I=1,17
    3 SR(I)=WR(I)*GFW(I)/228.35
C
C SR=MOLE RESPONSE FACTOR
C
      J=0.0
    4 J=J+1
      IF(J-M)5,5,24
    5 READ(5,6)N,D
    6 FORMAT(I2,9A8)
C
C N=NO OF MINOR PEAKS
C D=SAMPLE DESCRIPTION
C
      READ(5,7)(X(I),I=1,17)
    7 FORMAT(17F0.0)
C
C X=UNCORRECTED AREAS OF MAJOR PEAKS
C
      DO 8 I=1,17
    8 CORW(I)=X(I)/WR(I)
C
C CORW=MAJOR PEAK AREAS CORRECTED FOR WEIGHT RESPONSE
C
      DO 9 I=1,17
    9 CORM(I)=X(I)/SR(I)

C
C CORM=MAJOR PEAK AREAS CORRECTED FOR MOLE RESPONSE
C
      WTAR=0.0
C
C WTAR=SUM OF CORRECTED WEIGHT AREAS FOR MAJOR PEAKS
C
      DO 10 I=1,17
   10 WTAR=WTAR+CORW(I)
      SLAR=0.0
C
C SLAR=SUM OF CORRECTED MOLE AREAS FOR MAJOR PEAKS
C
      DO 11 I=1,17
   11 SLAR=SLAR+CORM(I)
      READ(5,12)(Y(I),I=1,N)
   12 FORMAT(20F0.0)
C
C Y=UNCORRECTED AREAS OF MINOR PEAKS
C
      SUMN=0.0
C
C SUMN=TOTAL AREA OF MINOR PEAKS
C
      DO 13 I=1,N
   13 SUMN=SUMN+Y(I)
      WTOT=SUMN+WTAR
C
C WTOT=TOTAL AREA OF ALL PEAKS (WEIGHT RESPONSE CORR)
C
```

```
      STOT=SUMN+SLAR
C
C STOT=TOTAL AREA OF ALL PEAKS (MOLE RESPONSE CORR)
C
      PWMJ=WTAR/WTOT*100.0
C
C PWMJ=PERCENT MAJOR PEAKS OF TOTAL WEIGHT CORR AREA
C
      PMMJ=SLAR/STOT*100.0
C
C PMMJ=PERCENT MAJOR PEAKS OF TOTAL MOLE CORR AREA
C
      PWMN=100.0-PWMJ
C
C PWMN=PERCENT MINOR PEAKS OF TOTAL WEIGHT CORR AREA
C
      PMMN=100.0-PMMJ
C
C PMMN=PERCENT MINOR PEAKS OF TOTAL MOLE CORR AREA
C
      DO 14 I=1,17
   14 WTPC(I)=CORW(I)/WTOT*100.0
C
C WTPC=CONCENTRATION IN WEIGHT PERCENT
C
      DO 15 I=1,17
   15 SLPC(I)=CORM(I)/STOT*100.0
C
C SLPC=CONCENTRATION IN MOLE PERCENT
C

      UNMJ =0.0
C
C UNMJ=TOTAL UNCORRECTED AREA OF MAJOR PEAKS
C
      DO 16 I=1,17
   16 UNMJ=UNMJ+X(I)
      UNTO=UNMJ+SUMN
C
C UNTO=TOTAL UNCORRECTED AREA
C
      WTUA=0.0
C
C WTUA=WEIGHT PERCENT TOTAL UNSAT ACIDS
C
      WTUA=WTPC(7)+WTPC(10)+WTPC(13)+WTPC(14)+WTPC(15)
      STUA=0.0
C
C STUA=MOLE PERCENT TOTAL UNSAT ACIDS
C
      STUA=SLPC(7)+SLPC(10)+SLPC(13)+SLPC(14)+SLPC(15)
      WTSA=PWMJ-WTUA
      STSA=PMMJ-STUA
C
C STSA=MOLE PERCENT TOTAL SAT ACIDS
C WTSA=WEIGHT PERCENT TOTAL SAT ACIDS
C
      WMIA=WTPC(14)+WTPC(15)
      SMIA=SLPC(14)+SLPC(15)
C
C WMIA=WEIGHT PERCENT CIS-METHYLENE INTERRUPTED ACIDS
C SMIA=MOLE PERCENT CIS-METHYLENE INTERRUPTED ACIDS
C
C GLCM=NO OF MOLES OF GLYCEROL ASSOC WITH MAJOR ACIDS
C GLMN=NO OF MOLES OF GLYCEROL ASSOC WITH MINOR ACIDS
C
      GLCM=0.0
      DO 17 I=1,17
   17 GLCM=GLCM+WTPC(I)/GFW(I)
      GLCM=GLCM/3.0
      GLMN=PWMN/(228.35+3.0)
```

```
C
C  GLMT=TOTAL NO OF MOLES OF GLYCEROL
C  GLCW=WEIGHT OF GLYCEROL ASSOC WITH 100 G OF FATTY ACIDS
C
      GLMT=GLCM+GLMN
      GLCW=GLMT*92.095
C
C  COGL=CONCENTRATION OF GLYCEROL IN FAT
C
      COGL=GLCW*100/(100+GLCW)
      UNIO=(WTPC(7)/226.35+WTPC(10)/254.39+WTPC(13)/282.46)*253.8+WTPC(1
     14)/280.43*507.6+WTPC(15)/278.41*761.4
C
C  UNIO=UNCORR IODINE VALUE
C
      COIO=UNIO*100/(100+GLCW)
C
C  COIO=CORRECTED IODINE NO
C
      WRITE(2,18)D,N

   18  FORMAT(1H1,50X,19HFATTY ACID ANALYSIS///10X,21HSAMPLE DESCRIPTION
     1= ,9A8//10X,21HNO OF MINOR PEAKS =  ,I2//27X,6HUNCORR,8X,6HWEIGHT,
     29X,6HWEIGHT,9X,4HCONC,10X,4HMOLE,11X,4HMOLE,9X,4HCONC/9X,10HFATTY
     3ACID,9X,4HAREA,8X,8HRESPONSE,6X,9HCORR AREA,6X,8HWEIGHT %,6X,8HRES
     4PONSE,6X,9HCORR AREA,6X,6HMOLE %)
      I=1
   19  WRITE(2,20)ANAME(1,I),ANAME(2,I),X(I),WR(I),CORW(I),WTPC(I),SR(I),
     1CORM(I),SLPC(I)
   20  FORMAT(1H0,6X,2A8,2X,F10.0,8X,F4.2,7X,F10.0,7X,F5.2,10X,F4.2,7X,F1
     10.0,6X,F5.2)
      IF(I-17)21,22,22
   21  I=I+1
      GO TO 19
   22  WRITE(2,23)UNMJ,WTAR,PWMJ,SLAR,PMMJ,SUMN,PWMN,PMMN,UNTO,WTOT,STOT,
     1WTSA,STSA,WTUA,STUA,WMIA,SMIA,COGL,COIO
   23  FORMAT(1H0/7X,16HMAJOR PEAKS      ,2X,F10.0,9X,1H-,9X,F10.0,7X,F5.2
     1,11X,1H-,9X,F10.0,6X,F5.2//7X,16HMINOR PEAKS      ,2X,F10.0,9X,1H-,
     214X,1H-,11X,F5.2,11X,1H-,14X,1H-,10X,F5.2//7X,16HTOTAL
     32X,F10.0,9X,1H-,9X,F10.0,6X,5H100.0,12X,1H-,9X,F10.0,5X,5H100.0///
     47X,21HSATURATED FATTY ACIDS,43X,F5.2,37X,F5.2//7X,23HUNSATURATED F
     5ATTY ACIDS,41X,F5.2,37X,F5.2//7X,47HPOLYUNSAT CIS-METHYLENE INTERR
     6UPTED FATTY ACIDS,17X,F5.2,37X,F5.2//7X,10HGLYCEROL %,20X,F5.2//7X
     7,21HTHEORETICAL IODINE NO,8X,F6.2)
      GO TO 4
   24  STOP
      END

END OF SEGMENT, LENGTH   708, NAME  FATAANALYSIS
```

DISCUSSION

The program and experimental procedure were designed primarily for the fatty acid analysis of butterfat and margarine fat. For other fats and oils, the range of acids, experimental conditions, and calculation methods may require alteration. The decision to use the butyl esters of the acids resulted from preliminary work which showed that with these esters no loss of the short-chain acid esters occurred and peak separation was excellent. In particular, butyric acid separated more readily from the solvent peak as its butyl ester than as its methyl ester.

To justify assigning weight and molar response factors of 1.0 to unidentified peaks, it is important to identify as many peaks as possible on the chromatogram and

Sample printout

FATTY ACID ANALYSIS

SAMPLE DESCRIPTION - MEDIUM FRACTION TRIGLYCERIDE BAND 9/1/76

NO OF MINOR PEAKS - 6

FATTY ACID		UNCORR AREA	WEIGHT RESPONSE	WEIGHT CORR AREA	CONC WEIGHT %	MOLE RESPONSE	MOLE CORR AREA	CONC MOLE %
YRIC	C 4.0	15000.	1.35	11111.	2.40	0.52	28799.	6.60
ROIC	C 6.0	7115.	1.22	5832.	1.26	0.52	11465.	2.63
RYLIC	C 8.0	3306.	1.12	2952.	0.64	0.71	4674.	1.07
RIC	C10.0	8273.	1.07	7732.	1.67	0.81	10249.	2.35
RIC	C12.0	11723.	1.01	11607.	2.51	0.89	13232.	3.03
ISTIC	C14.0	47577.	1.00	47577.	10.28	1.00	47577.	10.91
ISTOLEI	C14.1	4540.	1.00	4540.	0.98	0.99	4580.	1.05
NTADECAN	C15.0	8895.	1.00	8895.	1.92	1.06	8380.	1.92
LMITIC	C16.0	137683.	0.98	140493.	30.37	1.10	125118.	28.69
LMITOLEI	C16.1	7755.	0.96	8078.	1.75	1.37	7251.	1.66
RGARIC	C17.0	8988.	0.97	9266.	2.00	1.15	7824.	1.79
EARIC	C18.0	101459.	0.94	105686.	22.84	1.20	84840.	19.45
EIC	C18.1	75606.	0.94	80432.	17.38	1.16	65028.	14.91
NOLEIC	C18.2	1053.	0.79	1333.	0.29	0.97	1085.	0.25
NOLENIC	C18.3	0.	0.62	0.	0.00	0.76	0.	0.00
NADECANO	C19.0	3629.	0.96	3780.	0.82	1.25	2892.	0.66
ACHIDIC	C20.0	600.	0.94	638.	0.14	1.29	466.	0.11
JOR PEAKS		443202.	-	443952.	97.25	-	423462.	97.09
NOR PEAKS		12707.	-	-	2.75	-	-	2.91
TAL		455909.	-	462659.	100.0	-	436169.	100.0

TURATED FATTY ACIDS	79.85 / 79.22
NSATURATED FATTY ACIDS	20.40 / 17.87
LYUNSAT CIS-METHYLENE INTERRUPTED FATTY ACIDS	0.29 / 0.25
LYCEROL %	11.25
HEORETICAL IODINE NO	16.85

hence obtain their response factors from analysis of standards. In the case of butterfat and margarine fat, the minor peaks contribute less than 4% of the total area. The error due to response differences is therefore insignificant. Assigning values of 1.0 to both response factors for the minor peaks is equivalent to assuming that each has the molecular weight of C14:0. For butterfat, the minor peaks occur approximately symmetrically before and after the C14:0 peak.

Unsaturation in a fat or oil is often estimated as the experimentally determined iodine value. In order to make a direct comparison between a published iodine value and a fatty acid profile, it is useful to compute the theoretical iodine value from the fatty acid profile. In the case of butterfat or margarine fat, the theoretical iodine value is calculated as the weight of iodine that would react with the double bonds of myristoleic (C14:1), palmitoleic (C16:1), oleic (C18:1), linoleic (C18:2), and linolenic

(C18:3) acids in 100 g of fat or oil. At this stage, theoretical and experimental iodine values have been compared for only a few samples. The results obtained indicate that the correlation is high and further confirmation will be the subject of another report.

The procedure described has been used successfully in this laboratory for over twelve months. The only notable problems are the minor delays that occur during transcription of data onto coding forms and card punching. The advantages of the procedure are that more samples can be analysed with less operator involvement and fewer mathematical errors than by manual computation methods.

REFERENCES

1 L. M. Smith and S. Hardjo, *Lipids*, 9 (1974) 674.
2 P. W. Parodi, *Aust. J. Dairy Technol.*, 26 (1971) 155.
3 E. C. Don Santos, *Arq. Esc. Vet., Univ. Fed. Minas Gerais*, 25 (1973) 99.
4 R. Watts and R. Dils, *Lipids*, 3 (1968) 471.
5 M. S. F. Lie Ken Jie, *J. Chromatogr.*, 109 (1975) 81.
6 M. W. Taylor and J. C. Hawke, *N.Z. J. Dairy Sci. Technol.*, 10 (1975) 49.
7 P. W. Parodi, *J. Chromatogr.*, 111 (1975) 223.
8 S. W. Christopherson and R. L. Glass, *J. Dairy Sci.*, 52 (1969) 1289.